
§20.1 THE PLANE BAR ELEMENT

This Chapter begins Part III of the course. This Part deals with the computer implementation of
the Finite Element Method for static analysis. It is organized in “bottom up” fashion. It begins
with simple topics, such as programming of bar and beam elements, and gradually builds up toward
more complex models and calculations.

Specific examples of this Chapter illustrate the programming of one-dimensional elements: bars
and beams, using Mathematica as implementation language.

§20.1. The Plane Bar Element

The two-node, prismatic, two-dimensional bar element
was studied in Chapters 2-3 for modeling plane trusses.
It is reproduced in Figure 20.1 for conveniency. It has
two nodes and four degrees of freedom. The element
node displacements and conjugate forces are

 , fe =

 . (20.1)

Figure 20.1. Plane bar element.

The element geometry is described by the coordinates {xi , yi }, i = 1, 2 of the two end nodes. For
stiffness computations, the only material and fabrication properties required are the modulus of
elasticity E = Ee and the cross section area A = Ae, respectively. Both are taken to be constant
over the element.

§20.1.1. Plane Bar Stiffness Matrix

The element stiffness matrix in global {x, y} coordinates is given by the explicit expression derived
in §3.1:

Ke = E A

�




c2 sc −c2 −sc
sc s2 −sc −s2

−c2 −sc c2 sc
−sc −s2 sc s2


 = E A

�3




x21x21 x21 y21 −x21x21 −x21 y21

x21 y21 y21 y21 −x21 y21 −y21 y21

−x21x21 −x21 y21 x21x21 x21 y21

−x21 y21 −y21 y21 x21 y21 y21 y21


 . (20.2)

Here c = cos ϕ = x21/�, s = sin ϕ = y21/�, in which x21 = x2−x1, y21 = y2−y1, � =
√

x2
21 + y2

21,
and ϕ is the angle formed by x̄ and x , measured from x positive counterclockwise — see Figure 20.1.
The second expression in (20.2) is preferable in a computer algebra system because it enhances
simplification possibilities when doing symbolic work, and is the one actually implemented in the
module described below.

§20.1.2. Plane Bar Stiffness Module

The computation of the stiffness matrix Ke of the two-node, prismatic plane bar ele-
ment is done by Mathematica module PlaneBar2Stiffness. This is listed in Fig-
ure IFEM:Ch20:fig:PlaneBarStiffMatrixModule. The module is invoked as

Ke = PlaneBar2Stiffness[ncoor, Em, A, options] (20.3)

Chapter 20: IMPLEMENTATION OF ONE-DIMENSIONAL ELEMENTS

PlaneBar2Stiffness[ncoor_,Em_,A_,options_]:= Module[
 {x1,x2,y1,y2,x21,y21,EA,numer,L,LL,LLL,Ke}, 
  {{x1,y1},{x2,y2}}=ncoor; {x21,y21}={x2-x1,y2-y1};
  EA=Em*A; {numer}=options; LL=x21^2+y21^2; L=Sqrt[LL];
  If [numer,{x21,y21,EA,LL,L}=N[{x21,y21,EA,LL,L}]]; 
  If [!numer, L=PowerExpand[L]]; LLL=Simplify[LL*L];
  Ke=(Em*A/LLL)*{{ x21*x21, x21*y21,-x21*x21,-x21*y21},
                 { y21*x21, y21*y21,-y21*x21,-y21*y21},
                 {-x21*x21,-x21*y21, x21*x21, x21*y21},
                 {-y21*x21,-y21*y21, y21*x21, y21*y21}}; 
  Return[Ke]];   

Figure 20.2. Mathematica stiffness module for a two-node, prismatic plane bar element.

The arguments are

ncoor Node coordinates of element arranged as { { x1,y1 },{ x2,y2 } }.
Em Elastic modulus.
A Cross section area.
options A list of processing options. For this element is has only one entry: { numer }. This

is a logical flag with the value True or False. If True the computations are carried
out in floating-point arithmetic. If False symbolic processing is assumed.

The module returns the 4 × 4 element stiffness matrix as function value.

ClearAll[A,Em,L];
ncoor={{0,0},{30,40}}; Em=1000; A=5; 
Ke= PlaneBar2Stiffness[ncoor,Em,A,{True}];
Print["Numerical Elem Stiff Matrix: "];
Print[Ke//MatrixForm];
Print["Eigenvalues of Ke=",Chop[Eigenvalues[N[Ke]]]];
Print["Symmetry check=",Simplify[Chop[Transpose[Ke]-Ke]]];

36. 48. −36. −48.
48. 64. −48. −64.

−36. −48. 36. 48.
−48. −64. 48. 64.

Numerical Elem Stiff Matrix:

Eigenvalues of Ke = {200., 0, 0, 0}

Symmetry check={{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}}

Figure 20.3. Test of plane bar stiffness module with numerical inputs.

§20.1.3. Testing the Plane Bar Module

The modules are tested by the scripts listed in Figures 20.3 and 20.4. The script shown on the top
of Figure 20.3 tests a numerically defined element with end nodes located at (0, 0) and (30, 40),
with E = 1000, A = 5, and numer set to True. Executing the script produces the results listed in
the bottom of that figure.

§20.2 THE SPACE BAR ELEMENT

ClearAll[A,Em,L];
ncoor={{0,0},{L,0}};
Ke=  PlaneBar2Stiffness[ncoor,Em,A,{False}]; 
kfac=Em*A/L; Ke=Simplify[Ke/kfac];
Print["Symbolic Elem Stiff Matrix: "]; 
Print[kfac," ",Ke//MatrixForm];
Print["Eigenvalues of Ke=",kfac,"*",Eigenvalues[Ke]];

A Em
L

1 0 −1 0

0 0 0 0

−1 0 1 0

0 0 0 0

Symbolic Elem Stiff Matrix:

Eigenvalues of Ke =           ∗ {0, 0, 0, 2}A Em
   L

Figure 20.4. Test of plane bar stiffness module with symbolic inputs.

On return from PlaneBar2Stiffness, the stiffness matrix returned in Ke is printed. Its four
eigenvalues are computed and printed. As expected three eigenvalues, which correspond to the
three independent rigid body motions of the element, are zero. The remaining eigenvalue is positive
and equal to E A/�. The symmetry of Ke is checked by printing (Ke)T − Ke upon simplification
and chopping.

The script of Figure 20.4 tests a symbolically defined bar element with end nodes located at (0, 0)

and (L , 0), which is aligned with the x axis. Properties E and A are kept symbolic. Executing the
script shown in the top of Figure 20.4 produces the results shown in the bottom of that figure. One
thing to be noticed is the use of the stiffness scaling factor E A/�, called kfac in the script. This is
a symbolic quantity that can be extracted as factor of matrix Ke. The effect is to clean up matrix
and vector output, as can be observed in the printed results.

§20.2. The Space Bar Element

To show how the previous implementation extends easily to three dimensions, this section describes
the implementation of the space bar element.

The two-node, prismatic, space bar
element is pictured in Figure 20.5.
The element has two nodes and six
degrees of freedom. The element node
displacements and conjugate forces
are arranged as

(20.4)

Figure 20.5. The space (3D) bar element.

Chapter 20: IMPLEMENTATION OF ONE-DIMENSIONAL ELEMENTS

SpaceBar2Stiffness[ncoor_,Em_,A_,options_]:=Module[
 {x1,x2,y1,y2,z1,z2,x21,y21,z21,EA,numer,L,LL,LLL,Ke},
  {{x1,y1,z1},{x2,y2,z2}}=ncoor;{x21,y21,z21}={x2-x1,y2-y1,z2-z1};
  EA=Em*A; {numer}=options;  LL=x21^2+y21^2+z21^2; L=Sqrt[LL];
  If [numer,{x21,y21,z21,EA,LL,L}=N[{x21,y21,z21,EA,LL,L}]];
  If [!numer, L=PowerExpand[L]]; LLL=Simplify[LL*L];
  Ke=(Em*A/LLL)*
     {{ x21*x21, x21*y21, x21*z21,-x21*x21,-x21*y21,-x21*z21},
      { y21*x21, y21*y21, y21*z21,-y21*x21,-y21*y21,-y21*z21},
      { z21*x21, z21*y21, z21*z21,-z21*x21,-z21*y21,-z21*z21},
      {-x21*x21,-x21*y21,-x21*z21, x21*x21, x21*y21, x21*z21},
      {-y21*x21,-y21*y21,-y21*z21, y21*x21, y21*y21, y21*z21},
      {-z21*x21,-z21*y21,-z21*z21, z21*x21, z21*y21, z21*z21}};
   Return[Ke];
]; 

Figure 20.6. Module to form the stiffness of the space (3D) bar element.

The element geometry is described by the coordinates {xi , yi , zi }, i = 1, 2 of the two end nodes.
As in the case of the plane bar, the two properties required for the stiffness computations are the
modulus of elasticity E and the cross section area A. Both are assumed to be constant over the
element.

§20.2.1. SpaceBarStiffness Matrix

For the space bar element, introduce the notation x21 = x2 − x1, y21 = y2 − y1, z21 = z2 − z1 and

� =
√

x2
21 + y2

21 + z2
21. It can be shown1 that the element stiffness matrix in global coordinates is

given by

Ke = Ee Ae

�3




x21x21 x21 y21 x21z21 −x21x21 −x21 y21 −x21z21

x21 y21 y21 y21 x21z21 −x21 y21 −y21 y21 −y21z21

x21z21 y21z21 z21z21 −x21z21 −y21z21 −z21z21

−x21x21 −x21 y21 −x21z21 x21x21 x21 y21 x21z21

−x21 y21 −y21 y21 −x21z21 x21 y21 y21 y21 y21z21

−x21z21 −y21z21 −z21z21 x21z21 y21z21 z21z21




. (20.5)

This matrix expression in terms of coordinate differences is useful in symbolic work, because it
enhances simplification possibilities.

§20.2.2. Space Bar Stiffness Module

The computation of the stiffness matrix Ke of the two-node, prismatic space bar element, is done by
Mathematica module SpaceBar2Stiffness. This is listed in Figure 20.6. The module is invoked
as

Ke = SpaceBar2Stiffness[ncoor, Em, A, options] (20.6)

The arguments are

ncoor Node coordinates of element arranged as { { x1,y1,z1 },{ x2,y2,z2 } }.
1 The derivation was the subject of Exercise 6.10.

20–6



§20.2 THE SPACE BAR ELEMENT

ClearAll[A,Em];
ncoor={{0,0,0},{2,3,6}}; Em=343; A=10;
Ke= SpaceBar2Stiffness[ncoor,Em,A,{True}];
Print["Numerical Elem Stiff Matrix: "];
Print[Ke//MatrixForm];
Print["Eigenvalues of Ke=",Chop[Eigenvalues[Ke]]];

40. 60. 120. −40. −60. −120.
60. 90. 180. −60. −90. −180.

120. 180. 360. −120. −180. −360.
−40. −60. −120. 40. 60. 120.
−60. −90. −180. 60. 90. 180.

−120. −180. −360. 120. 180. 360.

Numerical Elem Stiff Matrix:

Eigenvalues of Ke = {980., 0, 0, 0, 0, 0}

Figure 20.7. Testing the space bar stiffness module with numerical inputs.

Em Elastic modulus.
A Cross section area.
options A list of processing options. For this element is has only one entry: { numer }. This

is a logical flag with the value True or False. If True the computations are carried
out in floating-point arithmetic. If False symbolic processing is assumed.

The module returns the 6 × 6 element stiffness matrix as function value.

ClearAll[A,Em,L];
ncoor={{0,0,0},{L,2*L,2*L}/3}; 
Ke=  SpaceBar2Stiffness[ncoor,Em,A,{False}];
kfac=Em*A/(9*L); Ke=Simplify[Ke/kfac];
Print["Symbolic Elem Stiff Matrix: "]; 
Print[kfac," ",Ke//MatrixForm];
Print["Eigenvalues of Ke=",kfac,"*",Eigenvalues[Ke]];

1 2 2 −1 −2 −2
2 4 4 −2 −4 −4
2 4 4 −2 −4 −4

−1 −2 −2 1 2 2
−2 −4 −4 2 4 4
−2 −4 −4 2 4 4

Symbolic Elem Stiff Matrix:

Eigenvalues of Ke =            ∗  {0, 0, 0, 0, 0, 18}
A Em
9 L

A Em
9 L

Figure 20.8. Testing the space bar stiffness module with symbolic inputs.

§20.2.3. Testing the Space Bar Module

The modules are tested by the scripts listed in Figures 20.7 and 20.8. As these are similar to previous
tests done on the plane bar they need not be described in detail.

Chapter 20: IMPLEMENTATION OF ONE-DIMENSIONAL ELEMENTS

Figure 20.9. Plane beam-column element: (a) in its local system; (b) in the global system.

The script of Figure 20.7 tests a numerically defined space bar with end nodes located at (0, 0, 0)

and (30, 40, 0), with E = 1000, A = 5 and numer set to True. Executing the script produces the
results listed in the bottom of that Figure.

The script of Figure 20.8 tests a symbolically defined bar element with end nodes located at (0, 0, 0)

and (L , 2L , 2L)/3, which has length L and is not aligned with the x axis. The element properties
E and A are kept symbolic. Executing the script produces the results shown in the bottom of that
Figure. Note the use of a stiffness factor kfac of E A/(9�) to get cleaner printouts.

§20.3. The Plane Beam-Column Element

Beam-column elements model structural members that resist both axial and bending actions. This
is the case in skeletal structures such as frameworks which are common in steel and reinforced-
concrete building construction. A plane beam-column element is a combination of a plane bar
(such as that considered in §20.1), and a plane beam.

We consider a beam-column element in its local system (x̄, ȳ) as shown in Figure 20.9(a), and then
in the global system (x, y) as shown in Figure 20.9(b). The six degrees of freedom and conjugate
node forces of the elements are:

ūx1

. (20.7)

The rotation angles θ and the nodal moments m are the same in the local and the global systems
because they are about the z axis, which does not change in passing from local to global.

The element geometry is described by the coordinates {xi , yi }, i = 1, 2 of the two end nodes. The
element length is � = Le. Properties involved in the stiffness calculations are: the modulus of
elasticity E , the cross section area A and the moment of inertia I = Izz about the neutral axis. All
properties are taken to be constant over the element.

§20.3.1. Plane Beam-Column Stiffness Matrix

§20.3 THE PLANE BEAM-COLUMN ELEMENT

To obtain the plane beam-column stiffness in the local system we simply add the stiffness matrices
derived in Chapters 11 and 12, respectively, to get

K̄
e = E A

�




1 0 0 −1 0 0
0 0 0 0 0

0 0 0 0
1 0 0

0 0
symm 0




+ E I

�3




0 0 0 0 0 0
12 6� 0 −12 6�

4�2 0 −6� 2�2

0 0 0
12 −6�

symm 4�2




(20.8)

The two matrices on the right of (20.8) come from the bar stiffness (12.22) and the Bernoulli-
Euler bending stiffness (13.20), respectively. Before adding them, rows and columns have been
rearranged in accordance with the nodal freedoms (20.7).

The displacement transformation matrix between local and global systems is

ūe =




ūx1

ū y1

θz1

ux2

ū y2

θz2




=




c s 0 0 0 0
−s c 0 0 0 0
0 0 1 0 0 0
0 0 0 c s 0
0 0 0 −s c 0
0 0 0 0 0 1







ux1

uy1

θz1

ux2

uy2

θz2




= T ue, (20.9)

in which c = cos ϕ = (x2 − x1)/�, s = sin ϕ = (y2 − y1)/�, and ϕ is the angle between x̄ and x ,
measured positive-counterclockwise from x ; see Figure IFEM:Ch20:fig:PlaneBeamColElem. The
stiffness matrix in the global system is obtained through the congruent transformation

Ke = TT K̄
e

T. (20.10)

Explicit expressions of the entries of Ke are messy. Unlike the bar, it is better to let the program do
the transformation.

§20.3.2. Plane Beam Column Stiffness Module

The computation of the stiffness matrix Ke of the two-node, prismatic plane beam-column element
is done by Mathematica module PlaneBeamColumn2Stiffness. This is listed in Figure 20.10.
The module is invoked as

Ke = PlaneBeamColumn2Stiffness[ncoor, Em, { A,Izz }, options] (20.11)

The arguments are

ncoor Node coordinates of element arranged as { { x1,y1 },{ x2,y2 } }.
Em Elastic modulus.
A Cross section area.
Izz Moment of inertia of cross section area respect to axis z.
options A list of processing options. For this element is has only one entry: { numer }. This

is a logical flag with the value True or False. If True the computations are carried
out in floating-point arithmetic. If False symbolic processing is assumed.

The module returns the 6 × 6 element stiffness matrix as function value.

Chapter 20: IMPLEMENTATION OF ONE-DIMENSIONAL ELEMENTS

PlaneBeamColumn2Stiffness[ncoor_,Em_,{A_,Izz_},options_]:= Module[
 {x1,x2,y1,y2,x21,y21,EA,EI,numer,L,LL,LLL,Te,Kebar,Ke}, 
  {{x1,y1},{x2,y2}}=ncoor; {x21,y21}={x2-x1,y2-y1}; 
   EA=Em*A; EI=Em*Izz; {numer}=options;
   LL=Simplify[x21^2+y21^2]; L=Sqrt[LL]; 
   If [numer,{x21,y21,EA,EI,LL,L}=N[{x21,y21,EA,EI,LL,L}]]; 
   If [!numer, L=PowerExpand[L]]; LLL=Simplify[LL*L]; 
   Kebar= (EA/L)*{
  { 1,0,0,-1,0,0},{0,0,0,0,0,0},{0,0,0,0,0,0},
  {-1,0,0, 1,0,0},{0,0,0,0,0,0},{0,0,0,0,0,0}} +
         (2*EI/LLL)*{
  { 0,0,0,0,0,0},{0, 6, 3*L,0,-6, 3*L},{0,3*L,2*LL,0,-3*L,  LL},
  { 0,0,0,0,0,0},{0,-6,-3*L,0, 6,-3*L},{0,3*L,  LL,0,-3*L,2*LL}};
   Te={{x21,y21,0,0,0,0}/L,{-y21,x21,0,0,0,0}/L,{0,0,1,0,0,0},
       {0,0,0,x21,y21,0}/L,{0,0,0,-y21,x21,0}/L,{0,0,0,0,0,1}};
   Ke=Transpose[Te].Kebar.Te; 
   Return[Ke] ];  

Figure 20.10. Mathematica module to form the stiffness matrix of a two-node,
prismatic plane beam-column element.

ClearAll[L,Em,A,Izz];
ncoor={{0,0},{3,4}}; Em=100; A=125; Izz=250; 
Ke= PlaneBeamColumn2Stiffness[ncoor,Em,{A,Izz},{True}];
Print["Numerical Elem Stiff Matrix: "];
Print[Ke//MatrixForm];
Print["Eigenvalues of Ke=",Chop[Eigenvalues[Ke]]];

2436. 48. −4800. −2436. −48. −4800.
48. 2464. 3600. −48. −2464. 3600.

−4800. 3600. 20000. 4800. −3600. 10000.
−2436. −48. 4800. 2436. 48. 4800.
−48. −2464. −3600. 48. 2464. −3600.

−4800. 3600. 10000. 4800. −3600. 20000.

Numerical Elem Stiff Matrix:

Eigenvalues of Ke = {34800., 10000., 5000., 0, 0, 0}

Figure 20.11. Test of two-node plane beam-column element with numeric inputs.

§20.3.3. Testing the Plane Beam-Column Module

The beam-column stiffness are tested by the scripts shown in Figures 20.11 and 20.12.

The script at the top of Figure 20.11 tests a numerically defined element of length � = 5 with
end nodes located at (0, 0) and (3, 4), respectively, with E = 100, A = 125 and Izz = 250. The
output is shown at the bottom of that figure. The stiffness matrix returned in Ke is printed. Its
six eigenvalues are computed and printed. As expected three eigenvalues, which correspond to the
three independent rigid body motions of the element, are zero. The remaining three eigenvalues
are positive.

The script at the top of Figure 20.12 tests a plane beam-column of length L with end nodes at (0, 0)

and (3L/5, 4L/5). The properties E , A and Izz are kept in symbolic form. The output is shown
at the bottom of that figure. The printed matrix looks complicated because bar and beam coupling

20–10



§20.4 *PLANE BEAM WITH OFFSET NODES

occurs when the element is not aligned with the global axes. The eigenvalues are obtained in closed
symbolic form, and their simplicity provides a good check that the transformation matrix (20.9) is
orthogonal. Three eigenvalues are exactly zero; one is associated with the axial (bar) stiffness and
two with the flexural (beam) stiffness.

ClearAll[L,Em,A,Izz];
ncoor={{0,0},{3*L/5,4*L/5}}; 
Ke=  PlaneBeamColumn2Stiffness[ncoor,Em,{A,Izz},{False}]; 
Print["Symbolic Elem Stiff Matrix:"]; kfac=Em; 
Ke=Simplify[Ke/kfac]; Print[kfac," ",Ke//MatrixForm];
Print["Eigenvalues of Ke=",kfac,"*",Eigenvalues[Ke]];

Em

3 ( 64 Izz +3 A L2 )
25 L3

12 ( −12 Izz +A L2 )
25 L3 − 24 Izz

5 L2 − 3 ( 64 Izz +3 A L2 )
25 L3 − 12 ( −12 Izz +A L2 )

25 L3 − 24 Izz
5 L2

12 ( −12 Izz +A L2 )
25 L3

4 ( 27 Izz +4 A L2 )
25 L3

18 Izz
5 L2 − 12 ( −12 Izz +A L2 )

25 L3 − 4 ( 27 Izz +4 A L2 )
25 L3

18 Izz
5 L2

− 24 Izz
5 L2

18 Izz
5 L2

4 Izz
L

24 Izz
5 L2 − 18 Izz

5 L2
2 Izz

L

− 3 ( 64 Izz +3 A L2 )
25 L3 − 12 ( −12 Izz +A L2 )

25 L3
24 Izz

5 L2
3 ( 64 Izz +3 A L2 )

25 L3
12 ( −12 Izz +A L2 )

25 L3
24 Izz

5 L2

−12 ( −12 Izz +A L2 )
25 L3 − 4 ( 27 Izz +4 A L2 )

25 L3 − 18 Izz
5 L2

12 ( −12 Izz +A L2 )
25 L3

4 ( 27 Izz +4 A L2 )
25 L3 − 18 Izz

5 L2

− 24 Izz
5 L2

18 Izz
5 L2

2 Izz
L

24 Izz
5 L2 − 18 Izz

5 L2
4 Izz

L

Figure 20.12. Test of two-node plane beam-column element with symbolic inputs.

§20.4. *Plane Beam With Offset Nodes

§20.4.1. Plate Reinforced With Edge Beams

Consider a plate reinforced with edge beams, as shown in Figure 20.13(a). The conventional placement of the
nodes is at the plate midsurface and beam longitudinal (centroidal) axis. But those element centered locations
do not coincide. To assemble the structure it is necessary to refer both the plate and beam stiffness equations
to common locations, because such equations are only written at nodes. We assume that those connection
nodes, or simply connectors, will be placed at the plate midsurface, as sketched in Figure 20.13(b). With that
choice there is no need to change the plate equations. The beam connectors have been moved, however, from
their original centroidal positions. For the beam these connectors are also known as offset nodes.

Structural configurations such as that of Figure 20.13(a) are common in aerospace, civil and mechanical
engineering when shells or plates are reinforced with eccentric stiffeners.

The process of moving the beam stiffness equation to the offset nodes is called offsetting. It relies on setting up
multifreedom constraints (MFC) between centered and offset node freedoms, and applying the master-slave
congruential transformation introduced in Chapter 8. For simplicity we discuss only this process assuming
that the beam of Figure 20.13(b) is a plane beam whose freedoms are to be moved upwards by a distance
d, which is positive if going upward from beam centroid. Freedoms at connection and centroidal nodes are
declared to be master and slaves, respectively. They are labeled as shown in Figure 20.13(c,d). The original
stiffness equations referred to centroidal (slave) freedoms are

Figure 20.13. Plane beam with nodes offset for a rigid-link connection to plate.

E

Le




A 0 0 −A 0 0
12Izzc

(Le)2
6Izzc

(Le)2 0 −12Izzc

(Le)2 − 6Izzc

(Le)2

4Izzc
Le 0 − 6Izzc

(Le)2
2Izzc
Le

A 0 0
12Izzc

(Le)2 − 6Izzc

(Le)2

symm 4Izzc
Le







ūx1s

ū y1s

θz1s

ūx2s

ū y2s

θz2s




=




f̄x1s

f̄ y1s

mz1s

f̄x2s

f̄ y2s

mz2s




, or Ke
c ue

c = fe
c, (20.12)

in which A is the beam cross section area while Izzc denotes the section moment of inertia with respect to the
centroidal axis z̄c.

§20.4.2. Rigid Link Transformation

Kinematic constraints between master and centroidal (slave) freedoms are obtained assuming that they are
connected by rigid links as pictured in Figure 20.13(c,d). This gives the centroidal(slave)-to-master transfor-
mation 



ūx1c

ū y1c

θz1c

ūx2c

ū y2c

θz2c


 =




1 0 d 0 0 0
0 1 0 0 0 0
0 0 1 0 0 0
0 0 0 1 0 d
0 0 0 0 1 0
0 0 0 0 0 1







ūx1m

ūy1m

θz1m

ūx2m

ūy2m

θz2m


 or ue

s = Tsm ue
m . (20.13)

The inverse transformation: Tmc = T−1
cm is obtained on replacing d with −d, as is physically obvious. The

modified stiffness equations are obtained by the congruential transformation: TT
cm Ke

c TT
cm = TT

cm fe
c = fe

m ,
which yields

E

Le




A 0 d −A 0 −d

0 12 Izzc

(Le)2
6 Izzc

Le 0 −12 Izzc

(Le)2
6 Izzc

Le

d 6 Izzc
Le 4 Izzc+A d2 −d −6 Izzc

Le 2 Izzc−A d2

−A 0 −d A 0 d

0 −12 Izzc

(Le)2 −6 Izzc
Le 0 12 Izzc

(Le)2 −6 Izzc
Le

−d 6 Izzc
Le 2 Izzc−A d2 d −6 Izzc

Le 4 Izzc+A d2







ūx1m

ūy1m

θz1m

ūx2m

ūy2m

θz2m




=




f̄x1m

f̄y1m

mz1m

f̄x2m

f̄y2m

mz2m




(20.14)

Note that the modified equations are still referred to the local system {x̄m, ȳm} pictured in Figure 20.13(c).
Prior to assembly they should be transformed to the global system {x, y} prior to assembly.

The foregoing transformation procedure has a flaw: for standard plate elements it will introduce compatibility
errors at the interface between plate and beam. This may cause the beam stiffness to be significantly under-
estimated. See the textbook by Cook et. al. [149] for an explanation, and references therein. The following
subsections describes a different scheme that builds Ke

m directly and cures that incompatibility.

§20.4.3. Direct Fabrication of Offset Beam Stiffness

This approach directly interpolates displacements and strains from master nodes placed at distance d from the
beam longitudinal (centroidal) axis, as pictured in Figure 20.14. As usual the isoparametric coordinate ξ along
the beam element varies from −1 at node 1 through +1 at node 2. The following cross section geometric
properties are defined for use below:

A =
∫

Ae

d A, Sz =
∫

Ae

ȳ d A = A d, Izzc =
∫

Ae

ȳ2
c d A, Izzm =

∫
Ae

ȳ2 d A = Izzc + A d2, (20.15)

The inplane displacements are expressed in term of the master freedoms at nodes 1m and 2m . Using the
Bernoulli-Euler model gives

[
ūxm

ū ym

]
=

[
Nux1 −ȳ

∂ Nuy1

∂ x̄ −ȳ ∂ Nθ z1
∂ x̄ Nux2 −ȳ

∂ Nuy2

∂ x̄ −ȳ ∂ Nθ z2
∂ x̄

0 Nuy1 Nθ z1 0 Nuy2 Nθ z2

]



ūx1m

ūy1m

θz1m

ūx2m

ūy2m

θz2m


 (20.16)

in which Nux1 = 1
2 (1 − ξ)/2, Nux2 = 1

2 (1 + ξ)/2, Nuy1 = 1
4 (1 − ξ)2(2 + ξ), Nθ z1 = 1

8 �(1 − ξ)2(1 + ξ),
Nuy2 = 1

4 (1 + ξ)2(2 − ξ), Nθ z2 = − 1
8 �(1 + ξ)2(1 − ξ) are the usual bar and beam shape functions, but here

referred to the offset axis x̄m .

The axial strain is exx = ∂ux/∂ x̄ and the strain energy U e = 1
2

∫
V e E e2

xx dV where dV = A dx̄=A. Carrying

Figure 20.14. Plane beam fabricated directly from offset master node freedoms.

out the integral and differentiating twice with respect to the degrees of freedom yields the stiffness matrix

Ke = E

Le




A 0 −A d −A 0 A d

0 12 (Izzc+A d2)

(Le)2
6 (Izzc+A d2)

Le 0 −12 (Izzc+A d2)

(Le)2
6 (Izzc+A d2)

Le

−A d 6 (Izzc+A d2)
Le 4 (Izzc+A d2) A d −6 (Izzc+A d2)

Le 2 (Izzc+A d2)

−A 0 A d A 0 −A d

0 −12 (Izzc+A d2)

(Le)2
−6 (Izzc+A d2)

Le 0 12 (Izzc+A d2)

(Le)2
−6 (Izzc+A d2)

Le

A d 6 (Izzc+A d2)
Le 2 (Izzc+A d2) −A d −6 (Izzc+A d2)

Le 4 (Izzc+A d2)




= E A

Le




1 0 −d −1 0 d

0
12 (r 2

G + d2)

(Le)2
6 (r 2

G + d2)

Le 0
−12 (r 2

G + d2)

(Le)2
6 (r 2

G + d2)

Le

−d
6 (r 2

G + d2)

Le 4 (r 2
G + d2) d

−6 (r 2
G + d2)

Le 2 (r 2
G + d2)

−1 0 d 1 0 −1 d

0
−12 (r 2

G + d2)

(Le)2
−6 (r 2

G + d2)

Le 0
12 (r 2

G + d2)

(Le)2
−6 (r 2

G + d2)

Le

d
6 (r 2

G + d2)

Le 2 (r 2
G + d2) −d

−6 (r 2
G + d2)

Le 4 (r 2
G + d2)




(20.17)

In the second form, r 2
G = Izzc/A is the squared radius of gyration of the cross section about z.

Comparing the first form of Ke in (20.17) with (20.14) shows that the bending terms are significantly different
if d 
= 0. These differences underscore the limitations of the rigid link assumption.

§20.5. *Layered Beam Configurations

Another application of rigid link constraints is to modeling of layered beams, also called composite beams
as well as beam stacks. These are beam-columns fabricated with beam layers that are linked to operate
collectively as a beam member. In this section we study how to form the stiffness equations of beam stacks
under the following assumptions:

1. The overall cross section of the beam member is rectangular and prismatic (that is, the cross section is
constant along the longitudinal direction).

2. Both the beam stack and each of its constituent layers are prismatic.

3. The layers are of homogeneous isotropic material. Layer material, however, may vary from layer to
layer.

The key modeling assumption is: is interlayer slip allosed or not? The two cases are studied next.

§20.5.1. Layered Beam With No Interlayer Slip
The main modeling constraint here is: if all layers are of the same material, the stiffness equations should
reduce to those of a homogenous beam. To discuss how to meet this requirement it is convenient to introduce
a beam template that separates the stiffness matrix into basic and higher order components. Consider a
homogeneous, isotropic prismatic beam column element with elastic modulus E , cross section area A and
cross section second moment of inertia Ixxc with respect to its neutral (centroidal) axis. The template form of
the stiffness matrix in the local system is

K̄
e = Ke

b + Ke
h =




Kb1 0 0 −Kb1 0 0
0 0 0 0 0 0
0 0 Kb2 0 0 −Kb2

−Kb1 0 0 Kb1 0 0
0 0 0 0 0 0
0 0 −Kb2 0 0 Kb2


 + βh




0 0 0 0 0 0
0 Kh3 Kh4 0 −Kh3 Kh4

0 Kh4 Kh5 0 −Kh4 Kh5

0 0 0 0 0 0
0 −Kh3 −Kh4 0 Kh3 −Kh4

0 Kh4 Kh5 0 −Kh4 Kh5


 (20.18)

in which Kb1 = E A/Le, Kb2 = E Izzc/Le, Kh3 = 12E Izzc/(Le)3, Kh4 = 6E Izzc/(Le)2 and Kh5 = 3E Izzc/Le.
Here βh is a free parameter that scales the higher order stiffness Kh . If βh = 1 we recover the standard beam
column stiffness (20.8). For a rectangular cross section of height H and width h, A = Hh and Izzc = H 3 h/12,
and the template (20.18) becomes

K̄
e = E h

Le




H 0 0 −H 0 0
0 0 0 0 0 0
0 0 1

12 H 3 0 0 − 1
12 H 3

−H 0 0 H 0 0
0 0 0 0 0 0
0 0 − 1

12 H 3 0 0 1
12 H 3


 + βh E H 3 h

4(Le)3




0 0 0 0 0 0
0 4 2Le 0 −4 2Le

0 2Le (Le)2 0 −2Le (Le)2

0 0 0 0 0 0
0 −4 −2Le 0 4 −2Le

0 2Le (Le)2 0 −2Le (Le)2


 (20.19)

Next, cut the foregoing beam into two identical layers of height Hk = H/2, where k = 1, 2 is used as layer
index. See Figure 20.15(b). The layers have area Ak = Hk h = H h/2 and self inertia Ixxk = H 3

k h/12 =
H 3 h/96. The layer stiffness matrices in template form are

K̄
e
k = E h

Le




Hk 0 0 −Hk 0 0
0 0 0 0 0 0
0 0 1

12 H 3
k 0 0 − 1

12 H 3
k

−Hk 0 0 Hk 0 0
0 0 0 0 0 0
0 0 − 1

12 H 3
k 0 0 1

12 H 3
k


+βhk E H 3

k h

4(Le)3




0 0 0 0 0 0
0 4 2Le 0 −4 2Le

0 2Le (Le)2 0 −2Le (Le)2

0 0 0 0 0 0
0 −4 −2Le 0 4 −2Le

0 2Le (Le)2 0 −2Le (Le)2


 , k = 1, 2.

(20.20)

Beacuse the layers are identical it is reasonable to assume the same higher order free parameter for both layers,
that is, βh1 = βh2 = βh . The offset distances from each layer to the centroid of the full beam to the centroid
of each layer are d1 = −H/4 and d2 = H/4. The rigid-link transformation matrices for (20.20) are the same
as those found in the previous Section:

T1 =




1 0 −H/4 0 0 0
0 1 0 0 0 0
0 0 1 0 0 0
0 0 0 1 0 −H/4
0 0 0 0 1 0
0 0 0 0 0 1


 , T2 =




1 0 H/4 0 0 0
0 1 0 0 0 0
0 0 1 0 0 0
0 0 0 1 0 H/4
0 0 0 0 1 0
0 0 0 0 0 1


 . (20.21)

Figure 20.15. Plane beam divided into identical “sticking” layers.

Transforming and adding layers contributions as K̄
e = TT

1 K̄
e
1T1 + TT

2 K̄
e
2T2 gives

K̄
e = E h

Le




H 0 0 −H 0 0
0 0 0 0 0 0
0 0 1

12 H 3 0 0 − 1
12 H 3

−H 0 0 H 0 0
0 0 0 0 0 0
0 0 − 1

12 H 3 0 0 1
12 H 3


 + βh E H 3 h

16(Le)3




0 0 0 0 0 0
0 4 2Le 0 −4 2Le

0 2Le (Le)2 0 −2Le (Le)2

0 0 0 0 0 0
0 −4 −2Le 0 4 −2Le

0 2Le (Le)2 0 −2Le (Le)2


 . (20.22)

This becomes identical to (20.19) if we set βh = 4.

Carrying out the same exercise for three identical layers of height H/3, as shown in Figure 20.15(c), yields

K̄
e = E h

Le




H 0 0 −H 0 0
0 0 0 0 0 0
0 0 1

12 H 3 0 0 − 1
12 H 3

−H 0 0 H 0 0
0 0 0 0 0 0
0 0 − 1

12 H 3 0 0 1
12 H 3


 + βh E H 3 h

36(Le)3




0 0 0 0 0 0
0 4 2Le 0 −4 2Le

0 2Le (Le)2 0 −2Le (Le)2

0 0 0 0 0 0
0 −4 −2Le 0 4 −2Le

0 2Le (Le)2 0 −2Le (Le)2


 , (20.23)

which becomes identical to (20.19) if we set βh = 9. It is not difficult to show that if the beam is divided into
N ≥ 2 layers of height H/N , the correct beam stiffness is recovered if we take βh = N 2.

If is not difficult to prove the following generalization. Suppose that the beam is cut into N layers of heights
Hk = γk H , k = 1, . . . N that satisfy

∑N
1 Hk = H or

∑N
1 γk = 1. To get the correct stiffness of the layered

beam take
βhk = 1/γ 2

k . (20.24)

For example, suppose that the but is divided into 3 layers of thicknesses H1 = H3 = H/4 and H2 = H/2.
Then pick βh1 = βh3 = 1/( 1

4 )2 = 16 and βh2 = 1/( 1
2 )2 = 4.

What is the interpretation of this boost? A spectral analysis of the combined stiffness shows that taking βhk = 1
lowers the rigidity associated with the antisymmetric bending mode of the element. But this mode is associated
with shear-slippage between layers. Boosting βhk as found above compensates exactly for this rigidity decay.

§20.5.2. Beam Stacks Allowing Interlayer Slip

Sandwich beam fabrication
longitudinal cut

z

y

L = 80 cm

x

Elastic modulus of facings = E   = 65 GPa
Elastic modulus of core ~ 0

f

Figure 20.16. Plane sandwich beam.

There are beam fabrications where layers can slip longitudinally past each other. One important example is
the sandwich beam illustrated in Figure 20.16. The beam is divided into three layers: 2 metal sheet facings
and a honeycomb core. The core stiffness can be neglected since its effective elastic modulus is very low
compared to the facings modulus E f . In addition, the facings are not longitudinally bonded since they are
separated by the weaker core. To form the beam stiffness by the rigid link method it is sufficient to form the
faces self-stiffness, which are identical, and the rigid-link transformation matrices:

Ke
k = E f

Le




A f 0 0 −A f 0 0

0
12 Izz f

(Le)2
6 Izz f

Le 0 −12 Izz f

(Le)2
6 Izz f

Le

0
6 Izz f

Le 4 Izz f −d −6 Izz f

Le 2 Izz f

−A f 0 0 A f 0 0

0 −12 Izz f

(Le)2 −6 Izz f

Le 0
12 Izz f

(Le)2 −6 Izz f

Le

0
6 Izz f

Le 2 Izz f 0 −6 Izz f

Le 4 Izz f




, Tk =




1 0 d f 0 0 0
0 1 0 0 0 0
0 0 1 0 0 0
0 0 0 1 0 d f

0 0 0 0 1 0
0 0 0 0 0 1


 , k = 1, 2.

(20.25)

where, in the notation of Figure 20.16, A f = bt f , Ixx f = bt3
f /12, d1 = −(c + t f )/2 and d2 = (c + t f )/2.

The stiffness of the sandwich beam is Ke = TT
1 K bolde

1 T1 + TT
2 K bolde

2 T2 into which the numerical values
given in Figure 20.16 may be inserted. There is no need to use here the template form and of adjusting the
higher order stiffness.

§20.6. The Space Beam Element

A second example in 3D is the general beam element shown in Figure 20.17. The element is
prismatic and has two end nodes: 1 and 2, placed at the centroid of the end cross sections.

These define the local x̄ axis as directed from
1 to 2. For simplicity the cross section will
be assumed to be doubly symmetric, as is the
case in commercial I and double-T profiles.
The principal moments of inertia are defined
by these symmetries. The local ȳ and z̄ axes
are aligned with the symmetry lines of the
cross section forming a RH system with x̄ .
Consequently the principal moments of in-
ertia are Iyy and Izz , the bars being omitted
for convenience.

The global coordinate system is {x, y, z}. To
define the orientation of {ȳ, z̄} with respect
to the global system, a third orientation node
3, which must not be colinear with 1–2, is
introduced. See Figure 20.17. Axis ȳ lies in
the 1–2–3 plane and z̄ is normal to 1–2–3.

x−

y−

z−

2 (x  ,y  ,z  )

y

z x

2 2 2

1 (x  ,y  ,z  )1 1 1

3 (x  ,y  ,z  )3 3 3

− −
Orientation node
defining plane x,y

Figure 20.17. The space (3D) beam element.

Six global DOF are defined at each node i : the 3 translations uxi , uyi , uzi and the 3 rotations θxi ,
θyi , θzi .

§20.6.1. Space Beam Stiffness Matrix

The element global node displacements and conjugate forces are arranged as

ue = [ ux1 uy1 uz1 θx1 θy1 θz1 ux2 uy2 uz2 θx2 θy2 θz2 ]T ,

fe = [ fx1 fy1 fz1 mx1 my1 mz1 fx2 fy2 fz2 mx2 my2 mz2 ]T .
(20.26)

The beam material is characterized by the elastic modulus E and the shear modulus G (the latter
appears in the torsional stiffness). Four cross section properties are needed: the cross section area
A, the moment of inertia J that characterizes torsional rigidity,2 and the two principal moments of
inertia Iyy and Izz taken with respect to ȳ and z̄, respectively. The length of the element is denoted
by L . The Bernoulli-Euler model is used; thus the effect of tranverse shear on the beam stiffness is
neglected.

To simplify the following expressions, define the following “rigidity” combinations by symbols:
Ra = E A/L , Rt = G J/L , Rb

y3 = E Iyy/L3, Rb
y2 = E Iyy/L2, Rb

y = E Iyy/L , Rb
z3 = E Izz/L3,

Rb
z2 = E Izz/L2, Rb

z = E Izz/L . Note that Ra is the axial rigidity, Rt the torsional rigidity, while

2 For circular and annular cross sections, J is the polar moment of inertia of the cross section wrt x̄ . For other sections J
has dimensions of (length)4 but must be calculated according to St. Venant’s theory of torsion, or approximate theories.

§20.6 THE SPACE BEAM ELEMENT

SpaceBeamColumn2Stiffness[ncoor_,{Em_,Gm_},{A_,Izz_,Iyy_,Jxx_},
  options_]:= Module[
 {x1,x2,y1,y2,z1,z2,x21,y21,z21,xm,ym,zm,x0,y0,z0,dx,dy,dz,
  EA,EIyy,EIzz,GJ,numer,ra,ry,ry2,ry3,rz,rz2,rz3,rx,
  L,LL,LLL,yL,txx,txy,txz,tyx,tyy,tyz,tzx,tzy,tzz,T,Kebar,Ke}, 
  {x1,y1,z1}=ncoor[[1]]; {x2,y2,z2}=ncoor[[2]];
  {x0,y0,z0}={xm,ym,zm}={x1+x2,y1+y2,z1+z2}/2;
  If [Length[ncoor]<=2,{x0,y0,z0}+={0,1,0}]; 
  If [Length[ncoor]==3,{x0,y0,z0}=ncoor[[3]] ];
  {x21,y21,z21}={x2-x1,y2-y1,z2-z1}; {numer}=options;
  EA=Em*A; EIzz=Em*Izz; EIyy=Em*Iyy; GJ=Gm*Jxx;
  LL=Simplify[x21^2+y21^2+z21^2]; L=Sqrt[LL];
  If [numer, {x21,y21,z21,EA,EIyy,EIzz,GJ,LL,L}=
           N[{x21,y21,z21,EA,EIyy,EIzz,GJ,LL,L}]]; 
  If [!numer, L=PowerExpand[L]]; LLL=Simplify[LL*L];   
  ra=EA/L; rx=GJ/L;
  ry=2*EIyy/L; ry2=6*EIyy/LL; ry3=12*EIyy/LLL;
  rz=2*EIzz/L; rz2=6*EIzz/LL; rz3=12*EIzz/LLL; 
  Kebar={
  { ra,   0,   0,  0,   0,   0, -ra,   0,   0,   0,   0,   0},
  {  0, rz3,   0,  0,   0, rz2,   0,-rz3,   0,   0,   0, rz2},
  {  0,   0, ry3,  0,-ry2,   0,   0,   0,-ry3,   0,-ry2,   0}, 
  {  0,   0,   0, rx,   0,   0,   0,   0,   0, -rx,   0,   0},
  {  0,   0,-ry2,  0,2*ry,   0,   0,   0, ry2,   0,  ry,   0},
  {  0, rz2,   0,  0,   0,2*rz,   0,-rz2,   0,   0,   0,  rz},  
  {-ra,   0,   0,  0,   0,   0,  ra,   0,   0,   0,   0,   0},
  {  0,-rz3,   0,  0,   0,-rz2,   0, rz3,   0,   0,   0,-rz2},
  {  0,   0,-ry3,  0, ry2,   0,   0,   0, ry3,   0, ry2,   0},    
  {  0,   0,   0,-rx,   0,   0,   0,   0,   0,  rx,   0,   0},
  {  0,   0,-ry2,  0,  ry,   0,   0,   0, ry2,   0,2*ry,   0},
  {  0, rz2,   0,  0,   0,  rz,   0,-rz2,   0,   0,   0,2*rz}};
  {dx,dy,dz}={x0-xm,y0-ym,z0-zm};If[numer,{dx,dy,dz}=N[{dx,dy,dz}]];
  tzx=dz*y21-dy*z21; tzy=dx*z21-dz*x21; tzz=dy*x21-dx*y21;
  zL=Sqrt[tzx^2+tzy^2+tzz^2]; 
  If [!numer,zL=Simplify[PowerExpand[zL]]];
  {tzx,tzy,tzz}={tzx,tzy,tzz}/zL;   {txx,txy,txz}={x21,y21,z21}/L; 
  tyx=tzy*txz-tzz*txy; tyy=tzz*txx-tzx*txz; tyz=tzx*txy-tzy*txx;
  Te={{txx,txy,txz,   0,  0,  0,   0,  0,  0,   0,  0,  0},
      {tyx,tyy,tyz,   0,  0,  0,   0,  0,  0,   0,  0,  0},
      {tzx,tzy,tzz,   0,  0,  0,   0,  0,  0,   0,  0,  0},
      {  0,  0,  0, txx,txy,txz,   0,  0,  0,   0,  0,  0},
      {  0,  0,  0, tyx,tyy,tyz,   0,  0,  0,   0,  0,  0},
      {  0,  0,  0, tzx,tzy,tzz,   0,  0,  0,   0,  0,  0},
      {  0,  0,  0,   0,  0,  0, txx,txy,txz,   0,  0,  0},
      {  0,  0,  0,   0,  0,  0, tyx,tyy,tyz,   0,  0,  0},
      {  0,  0,  0,   0,  0,  0, tzx,tzy,tzz,   0,  0,  0},
      {  0,  0,  0,   0,  0,  0,   0,  0,  0, txx,txy,txz},
      {  0,  0,  0,   0,  0,  0,   0,  0,  0, tyx,tyy,tyz},
      {  0,  0,  0,   0,  0,  0,   0,  0,  0, tzx,tzy,tzz}};
   Ke=Transpose[Te].Kebar.Te; 
   Return[Ke]
 ];  

Figure 20.18. Module to form stiffness of space (3D) beam.

the Rb’s are bending rigities scaled by the length in various ways. Then the 12 × 12 local stiffness
matrix can be written as3

3 Cf. page 79 of Pzremieniecki [603]. The presentation in this book includes transverse shear effects as per Timoshenko’s
beam theory. The form (20.27) results from neglecting those effects.

K̄
e =




Ra 0 0 0 0 0 −Ra 0 0 0 0 0
0 12Rb

z3 0 0 0 6Rb
z2 0 −12Rb

z3 0 0 0 6Rb
z2

0 0 12Rb
y3 0 −6Rb

y2 0 0 0 −12Rb
y3 0 −6Rb

y2 0

0 0 0 Rt 0 0 0 0 0 −Rt 0 0
0 0 −6Rb

y2 0 4Rb
y 0 0 0 6Rb

y2 0 2Rb
y 0

0 6Rb
z2 0 0 0 4Rb

z 0 −6Rb
z2 0 0 0 2Rb

z
−Ra 0 0 0 0 0 Ra 0 0 0 0 0

0 −12Rb
z3 0 0 0 −6Rb

z2 0 12Rb
z3 0 0 0 −6Rb

z2
0 0 −12Rb

y3 0 6Rb
y2 0 0 0 12Rb

y3 0 6Rb
y2 0

0 0 0 −Rt 0 0 0 0 0 Rt 0 0
0 0 −6Rb

y2 0 2Rb
y 0 0 0 6Rb

y2 0 4Rb
y 0

0 6Rb
z2 0 0 0 2Rb

z 0 −6Rb
z2 0 0 0 4Rb

z




(20.27)

The transformation to the global system is the subject of Exercise 20.8.

§20.6.2. Space Beam Stiffness Module

The computation of the stiffness matrix Ke of the two-node, prismatic space beam-column element
is done by Mathematica module SpaceBeamColumn2Stiffness. This is listed in Figure 20.18.
The module is invoked as

Ke = SpaceBeamColumn2Stiffness[ncoor, { Em,Gm }, { A,Izz,Iyy,Jxx }, options]
(20.28)

The arguments are

ncoor Node coordinates of element arranged as { { x1,y1,z1 },{ x2,y2,z2 },{ x3,y3,z3 } },
in which { x3,y3,z3 } specifies an orientation node 3 that defines the local frame.
See §20.4.1.

Em Elastic modulus.
Gm Shear modulus.
A Cross section area.
Izz Moment of inertia of cross section area respect to axis z̄.
Iyy Moment of inertia of cross section area respect to axis ȳ.
Jxx Inertia with respect to x̄ that appears in torsional rigidity G J .
options A list of processing options. For this element is has only one entry: { numer }. This

is a logical flag with the value True or False. If True the computations are carried
out in floating-point arithmetic. If False symbolic processing is assumed.

The module returns the 12 × 12 element stiffness matrix as function value.

The implementation logic and testing of this element is the subject of Exercises 20.8 and 20.9.

Notes and Bibliography

All elements implemented here are formulated in most books dealing with matrix structural analysis. Przemie-
niecki [603] has been recommended in Chapter 1 on account of being inexpensive. Implementation and testing
procedures are rarely covered.

The use of rigid links for offsetting degrees of freedom is briefly covered in §7.8 of the textbook [149].

Exercises

Homework Exercises for Chapter 20

Implementation of One-Dimensional Elements

EXERCISE 20.1 [C:15] Download the plane bar stiffness module and their testers and verify the test results
reported here. Comment on whether the stiffness matrix Ke has the correct rank of 1.

EXERCISE 20.2 [C:15] Download the space bar stiffness module and their testers and verify the test results
reported here. Comment on whether the computed stiffness matrix Ke has the correct rank of 1.

EXERCISE 20.3 [C:15] Download the plane beam-column stiffness module and their testers and verify the
test results reported here. Comment on whether the computed stiffness matrix Ke has the correct rank of 3.

EXERCISE 20.4 [A+C:30] Explain why the space bar element has rank 1 although it has 6 degrees of freedom
and 6 rigid body modes. (According to the formula given in Chapter 19, the correct rank should be 6−6 = 0.)

EXERCISE 20.5 [C:25] Implement the plane bar, plane beam-column and space bar stiffness element module
in a lower level programming language and check them by writing a short test driver. [Do not bother about
the mass modules.] Your choices are C, Fortran 77 or Fortran 90. (C++ is overkill for this kind of software).

EXERCISE 20.6 [A:25] Explain why the eigenvalues of Ke of any the elements given here do not change if
the {x, y, z} global axes change.

EXERCISE 20.7 [A+C:30] (Advanced) Implement a 3-node space bar element. Hint: use the results of
Exercise 16.5 and transform the local stiffness to global coordinates via a 3 × 9 transformation matrix. Test
the element and verify that it has two nonzero eigenvalues.

EXERCISE 20.8 [D+A:25] Explain the logic of the space beam module listed in Figure 20.18. Assume that
the local stiffness matrix K̄

e
stored in Kebar is correct (it has been trranscribed from [603]. Instead, focus on

how the local to global transformation is built and applied.

EXERCISE 20.9 [C:25] Test the space beam element of Figure 20.18 using the scripts given in Figures E20.1
and E20.2, and report results. Comment on whether the element exhibits the correct rank of 6.

ClearAll[L,Em,Gm,A,Izz,Iyy,Jxx];
ncoor={{0,0,0},{1,8,4}}; Em=54; Gm=30; 
A=18; Izz=36; Iyy=72; Jxx=27;     
Ke= SpaceBeamColumn2Stiffness[ncoor,{Em,Gm},{A,Izz,Iyy,Jxx},{True}];
Print["Numerical Elem Stiff Matrix: "];
Print[SetPrecision[Ke,4]//MatrixForm];
Print["Eigenvalues of Ke=",Chop[Eigenvalues[Ke]]];

Figure E20.1. Script for numeric testing of the space beam module of Figure 20.18.

ClearAll[L,Em,Gm,A,Izz,Iyy,Jxx];
ncoor={{0,0,0},{2*L,2*L,L}/3}; 
Ke=SpaceBeamColumn2Stiffness[ncoor,{Em,Gm},{A,Izz,Iyy,Jxx},{False}];
kfac=Em; Ke=Simplify[Ke/kfac];
Print["Numerical Elem Stiff Matrix: "];
Print[kfac," ",Ke//MatrixForm];
Print["Eigenvalues of Ke=",kfac,"*",Eigenvalues[Ke]];

Figure E20.2. Script for symbolic testing of the space beam module of Figure 20.18.

